# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libxp
pkgver=1.0.3
pkgrel=2
pkgdesc="X11 X Print Library"
arch=('x86_64')
license=('custom')
url="https://xorg.freedesktop.org/"
depends=('libxext' 'printproto')
makedepends=('xorg-util-macros')
source=(${url}/releases/individual/lib/libXp-${pkgver}.tar.bz2{,.sig})
license=('custom')
sha512sums=('502a2500ec0f519dae69ac035bf93c65c47e3bc301604aacb350e90043455f3485626194db45208e5521300b282843825e08b7f7c412a7497ffa02f5f1560c8d'
            'SKIP')
validpgpkeys=('3BB639E56F861FA2E86505690FDD682D974CA72A') #  "Matt Turner <mattst88@gmail.com>"

build() {
  cd "${srcdir}/libXp-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/libXp-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
